// SPDX-License-Identifier: UNLICENSED

pragma solidity ^0.7.0;


// 
/**
 * @dev Wrappers over Solidity's arithmetic operations with added overflow
 * checks.
 *
 * Arithmetic operations in Solidity wrap on overflow. This can easily result
 * in bugs, because programmers usually assume that an overflow raises an
 * error, which is the standard behavior in high level programming languages.
 * `SafeMath` restores this intuition by reverting the transaction when an
 * operation overflows.
 *
 * Using this library instead of the unchecked operations eliminates an entire
 * class of bugs, so it's recommended to use it always.
 */
library SafeMath {
    /**
     * @dev Returns the addition of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `+` operator.
     *
     * Requirements:
     *
     * - Addition cannot overflow.
     */
    function add(uint256 a, uint256 b) internal pure returns (uint256) {
        uint256 c = a + b;
        require(c >= a, "SafeMath: addition overflow");

        return c;
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b) internal pure returns (uint256) {
        return sub(a, b, "SafeMath: subtraction overflow");
    }

    /**
     * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
     * overflow (when the result is negative).
     *
     * Counterpart to Solidity's `-` operator.
     *
     * Requirements:
     *
     * - Subtraction cannot overflow.
     */
    function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b <= a, errorMessage);
        uint256 c = a - b;

        return c;
    }

    /**
     * @dev Returns the multiplication of two unsigned integers, reverting on
     * overflow.
     *
     * Counterpart to Solidity's `*` operator.
     *
     * Requirements:
     *
     * - Multiplication cannot overflow.
     */
    function mul(uint256 a, uint256 b) internal pure returns (uint256) {
        // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
        // benefit is lost if 'b' is also tested.
        // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
        if (a == 0) {
            return 0;
        }

        uint256 c = a * b;
        require(c / a == b, "SafeMath: multiplication overflow");

        return c;
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b) internal pure returns (uint256) {
        return div(a, b, "SafeMath: division by zero");
    }

    /**
     * @dev Returns the integer division of two unsigned integers. Reverts with custom message on
     * division by zero. The result is rounded towards zero.
     *
     * Counterpart to Solidity's `/` operator. Note: this function uses a
     * `revert` opcode (which leaves remaining gas untouched) while Solidity
     * uses an invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b > 0, errorMessage);
        uint256 c = a / b;
        // assert(a == b * c + a % b); // There is no case in which this doesn't hold

        return c;
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b) internal pure returns (uint256) {
        return mod(a, b, "SafeMath: modulo by zero");
    }

    /**
     * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
     * Reverts with custom message when dividing by zero.
     *
     * Counterpart to Solidity's `%` operator. This function uses a `revert`
     * opcode (which leaves remaining gas untouched) while Solidity uses an
     * invalid opcode to revert (consuming all remaining gas).
     *
     * Requirements:
     *
     * - The divisor cannot be zero.
     */
    function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
        require(b != 0, errorMessage);
        return a % b;
    }
}

// 
/**
 * @dev Library for managing
 * https://en.wikipedia.org/wiki/Set_(abstract_data_type)[sets] of primitive
 * types.
 *
 * Sets have the following properties:
 *
 * - Elements are added, removed, and checked for existence in constant time
 * (O(1)).
 * - Elements are enumerated in O(n). No guarantees are made on the ordering.
 *
 * ```
 * contract Example {
 *     // Add the library methods
 *     using EnumerableSet for EnumerableSet.AddressSet;
 *
 *     // Declare a set state variable
 *     EnumerableSet.AddressSet private mySet;
 * }
 * ```
 *
 * As of v3.3.0, sets of type `bytes32` (`Bytes32Set`), `address` (`AddressSet`)
 * and `uint256` (`UintSet`) are supported.
 */
library EnumerableSet {
    // To implement this library for multiple types with as little code
    // repetition as possible, we write it in terms of a generic Set type with
    // bytes32 values.
    // The Set implementation uses private functions, and user-facing
    // implementations (such as AddressSet) are just wrappers around the
    // underlying Set.
    // This means that we can only create new EnumerableSets for types that fit
    // in bytes32.

    struct Set {
        // Storage of set values
        bytes32[] _values;

        // Position of the value in the `values` array, plus 1 because index 0
        // means a value is not in the set.
        mapping (bytes32 => uint256) _indexes;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function _add(Set storage set, bytes32 value) private returns (bool) {
        if (!_contains(set, value)) {
            set._values.push(value);
            // The value is stored at length-1, but we add 1 to all indexes
            // and use 0 as a sentinel value
            set._indexes[value] = set._values.length;
            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function _remove(Set storage set, bytes32 value) private returns (bool) {
        // We read and store the value's index to prevent multiple reads from the same storage slot
        uint256 valueIndex = set._indexes[value];

        if (valueIndex != 0) { // Equivalent to contains(set, value)
            // To delete an element from the _values array in O(1), we swap the element to delete with the last one in
            // the array, and then remove the last element (sometimes called as 'swap and pop').
            // This modifies the order of the array, as noted in {at}.

            uint256 toDeleteIndex = valueIndex - 1;
            uint256 lastIndex = set._values.length - 1;

            // When the value to delete is the last one, the swap operation is unnecessary. However, since this occurs
            // so rarely, we still do the swap anyway to avoid the gas cost of adding an 'if' statement.

            bytes32 lastvalue = set._values[lastIndex];

            // Move the last value to the index where the value to delete is
            set._values[toDeleteIndex] = lastvalue;
            // Update the index for the moved value
            set._indexes[lastvalue] = toDeleteIndex + 1; // All indexes are 1-based

            // Delete the slot where the moved value was stored
            set._values.pop();

            // Delete the index for the deleted slot
            delete set._indexes[value];

            return true;
        } else {
            return false;
        }
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function _contains(Set storage set, bytes32 value) private view returns (bool) {
        return set._indexes[value] != 0;
    }

    /**
     * @dev Returns the number of values on the set. O(1).
     */
    function _length(Set storage set) private view returns (uint256) {
        return set._values.length;
    }

   /**
    * @dev Returns the value stored at position `index` in the set. O(1).
    *
    * Note that there are no guarantees on the ordering of values inside the
    * array, and it may change when more values are added or removed.
    *
    * Requirements:
    *
    * - `index` must be strictly less than {length}.
    */
    function _at(Set storage set, uint256 index) private view returns (bytes32) {
        require(set._values.length > index, "EnumerableSet: index out of bounds");
        return set._values[index];
    }

    // Bytes32Set

    struct Bytes32Set {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _add(set._inner, value);
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(Bytes32Set storage set, bytes32 value) internal returns (bool) {
        return _remove(set._inner, value);
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(Bytes32Set storage set, bytes32 value) internal view returns (bool) {
        return _contains(set._inner, value);
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(Bytes32Set storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

   /**
    * @dev Returns the value stored at position `index` in the set. O(1).
    *
    * Note that there are no guarantees on the ordering of values inside the
    * array, and it may change when more values are added or removed.
    *
    * Requirements:
    *
    * - `index` must be strictly less than {length}.
    */
    function at(Bytes32Set storage set, uint256 index) internal view returns (bytes32) {
        return _at(set._inner, index);
    }

    // AddressSet

    struct AddressSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(AddressSet storage set, address value) internal returns (bool) {
        return _add(set._inner, bytes32(uint256(value)));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(AddressSet storage set, address value) internal returns (bool) {
        return _remove(set._inner, bytes32(uint256(value)));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(AddressSet storage set, address value) internal view returns (bool) {
        return _contains(set._inner, bytes32(uint256(value)));
    }

    /**
     * @dev Returns the number of values in the set. O(1).
     */
    function length(AddressSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

   /**
    * @dev Returns the value stored at position `index` in the set. O(1).
    *
    * Note that there are no guarantees on the ordering of values inside the
    * array, and it may change when more values are added or removed.
    *
    * Requirements:
    *
    * - `index` must be strictly less than {length}.
    */
    function at(AddressSet storage set, uint256 index) internal view returns (address) {
        return address(uint256(_at(set._inner, index)));
    }


    // UintSet

    struct UintSet {
        Set _inner;
    }

    /**
     * @dev Add a value to a set. O(1).
     *
     * Returns true if the value was added to the set, that is if it was not
     * already present.
     */
    function add(UintSet storage set, uint256 value) internal returns (bool) {
        return _add(set._inner, bytes32(value));
    }

    /**
     * @dev Removes a value from a set. O(1).
     *
     * Returns true if the value was removed from the set, that is if it was
     * present.
     */
    function remove(UintSet storage set, uint256 value) internal returns (bool) {
        return _remove(set._inner, bytes32(value));
    }

    /**
     * @dev Returns true if the value is in the set. O(1).
     */
    function contains(UintSet storage set, uint256 value) internal view returns (bool) {
        return _contains(set._inner, bytes32(value));
    }

    /**
     * @dev Returns the number of values on the set. O(1).
     */
    function length(UintSet storage set) internal view returns (uint256) {
        return _length(set._inner);
    }

   /**
    * @dev Returns the value stored at position `index` in the set. O(1).
    *
    * Note that there are no guarantees on the ordering of values inside the
    * array, and it may change when more values are added or removed.
    *
    * Requirements:
    *
    * - `index` must be strictly less than {length}.
    */
    function at(UintSet storage set, uint256 index) internal view returns (uint256) {
        return uint256(_at(set._inner, index));
    }
}

// 
interface IHornLockVault {
    function balanceOf(address account) external view returns (uint256);
    function claimableBalanceOf(address account) external view returns (uint256);
    function claimableHornOf(address account) external view returns (uint256);
    function deposit(uint256 amount, address referralAddr) external payable returns(bool);
    function withdraw() external payable returns(bool);
    function burn(uint256 amount) external payable returns(bool);
    function reserve() external view returns (uint256);
    function lockedAssets() external view returns (uint256);
    
    event Deposit(address indexed from, uint256 value, uint256 fee, uint256 fromDate, uint256 toDate);
    event Withdraw(address indexed to, uint256 value);
    event Burn(address indexed from, uint256 value, uint256 weightAdded);
    event DepositReward(address indexed to, uint256 reward, uint256 totalFee, uint256 senderBalance, uint256 contractBalance, uint256 weightPercent);
    event ReferralReward(address indexed from, address indexed to, uint256 amount, uint256 reward, uint256 baseFee);
    event NewPool(uint256 index, uint256 timestamp);
    event Claim(address indexed from, uint256 amount);
    event Log(string text);
    event LogUINT(uint256 value);
    event LogUINTText(string key, uint256 value);
    event LogAddress(address value);
}

// 
/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IExtendedERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    function symbol() external view returns (string memory);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);


    function mint(address to, uint256 amount) external;
    function burn(address from, uint256 amount) external;

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// 
contract HornLockVaultV2 is IHornLockVault {
    using SafeMath for uint256;

    struct LockedAsset {
        address account;
        uint256 amount;
        uint256 burnedHorn;
        uint256 fees;
        uint256 fromDate;
        uint256 toDate;
        uint256 alreadyClaimedHorn;
        bool isBurnAsset;
        uint256 depositIndex;
        uint256 enterPoolFees;
    }

    struct RewardFee {
        uint256 totalAssetAmount;
        uint256 totalFeesAmount;
        uint256 amount;
        uint256 depositIndex;
    }

    mapping(address => LockedAsset[]) private _lockedAssetsByAddress;
    mapping(uint256 => uint256) private _feesAtIndex;
    RewardFee private _poolFees;

    bool public _isActive = true;
    address public _createdFrom;
    address public _owner;
    string public _vaultName;
    uint128 public _fee;
    uint128 public _depositRewardFee;
    uint256 public _hornPerDay;
    uint256 public _feesVault;
    uint256 public _minLockDays;
    uint256 public _weightPerHorn;
    uint256 private _depositIndex = 0;
    bool public _isPaused = false;
    bool public _hornRewardDisabled = false;
    IExtendedERC20 private _token;
    address public _tokenAddr;
    IExtendedERC20 private _hornToken;
    address public _hornTokenAddr;

    constructor(
        address owner,
        address lockedTokenAddr,
        address hornTokenAddr,
        uint128 fee,
        uint128 depositRewardFee,
        uint256 hornPerDay,
        uint256 minLockDays,
        uint256 weightPerHorn
    ) {
        _createdFrom = msg.sender;
        _owner = owner;
        _tokenAddr = lockedTokenAddr;
        _token = IExtendedERC20(lockedTokenAddr);
        _hornTokenAddr = hornTokenAddr;
        _hornToken = IExtendedERC20(hornTokenAddr);
        _vaultName = _token.symbol();
        _fee = fee;
        _depositRewardFee = depositRewardFee;
        _hornPerDay = hornPerDay;
        _minLockDays = minLockDays;
        _weightPerHorn = weightPerHorn;

        _poolFees = RewardFee({
            amount: 0,
            totalFeesAmount: 0,
            totalAssetAmount: 0,
            depositIndex: _depositIndex
        });
    }

    function balanceOf(address account) public view override returns (uint256) {
        uint256 totalBalance = 0;
        for (uint256 i = 0; i < _lockedAssetsByAddress[account].length; i++) {
            if (_lockedAssetsByAddress[account][i].isBurnAsset || _lockedAssetsByAddress[account][i].amount <= 0) continue;
            totalBalance = totalBalance.add(
                _lockedAssetsByAddress[account][i].amount
            );
        }
        return totalBalance;
    }

    function claimableBalanceOf(address account)
        public
        view
        override
        returns (uint256)
    {
        uint256 totalBalance = 0;
        for (uint256 i = 0; i < _lockedAssetsByAddress[account].length; i++) {
            if (_lockedAssetsByAddress[account][i].isBurnAsset) continue;
            if (_lockedAssetsByAddress[account][i].toDate <= block.timestamp) {
                totalBalance = totalBalance.add(
                    _lockedAssetsByAddress[account][i].amount
                );
            }
        }
        return totalBalance;
    }

    function claimableHornOf(address account)
        public
        view
        override
        returns (uint256)
    {
        uint256 totalBalance = 0;
        for (uint256 i = 0; i < _lockedAssetsByAddress[account].length; i++) {
            if (
                _lockedAssetsByAddress[account][i].amount > 0 &&
                !_lockedAssetsByAddress[account][i].isBurnAsset
            ) {
                LockedAsset memory asset = _lockedAssetsByAddress[account][i];
                uint256 period =
                    block.timestamp.sub(asset.fromDate).div(60).div(60).div(24);
                if (period <= 0) continue;
                uint256 reward =
                    _hornPerDay.mul(asset.amount.mul(100)).div(10000).mul(
                        period
                    );
                reward = reward.div(100).sub(asset.alreadyClaimedHorn);
                totalBalance = totalBalance.add(reward);
            }
        }
        return totalBalance;
    }

    function reserve() public view override returns (uint256) {
        return _hornToken.balanceOf(address(this));
    }

    function lockedAssets() public view override returns (uint256) {
        uint256 totalBalance = _token.balanceOf(address(this)).sub(_feesVault);
        return totalBalance;
    }

    function getBurnedHornAmount(address account)
        public
        view
        returns (uint256)
    {
        uint256 totalBalance = 0;
        for (uint256 i = 0; i < _lockedAssetsByAddress[account].length; i++) {
            if (
                account == _lockedAssetsByAddress[account][i].account &&
                _lockedAssetsByAddress[account][i].isBurnAsset
            ) {
                totalBalance = totalBalance.add(
                    _lockedAssetsByAddress[account][i].burnedHorn
                );
            }
        }
        return totalBalance;
    }

    function name() public view returns (string memory) {
        return _vaultName;
    }

    function setPauseState(bool newState) external payable returns (bool) {
        require(
            msg.sender == _owner || msg.sender == _createdFrom,
            "Only the owner can do this"
        );
        _isPaused = newState;
        return true;
    }

    function setActiveState(bool newState) external payable returns (bool) {
        require(
            msg.sender == _owner || msg.sender == _createdFrom,
            "Only the owner can do this"
        );
        _isActive = newState;
        return true;
    }

    function setHornRewardDisabledState(bool newState)
        external
        payable
        returns (bool)
    {
        require(
            msg.sender == _owner || msg.sender == _createdFrom,
            "Only the owner can do this"
        );
        _hornRewardDisabled = newState;
        return true;
    }

    function deposit(uint256 amount, address referralAddr)
        external
        payable
        override
        returns (bool)
    {
        require(amount > 0, "Invalid amount");
        require(!_isPaused, "Deposit are paused for now");
        _deposit(msg.sender, amount, referralAddr);
        return true;
    }

    function balanceIndexes(address account, bool filterBurned)
        public
        view
        returns (uint256[] memory)
    {
        uint256 count = 0;
        for (uint256 i = 0; i < _lockedAssetsByAddress[account].length; i++) {
            if (
                _lockedAssetsByAddress[account][i].amount > 0 &&
                _lockedAssetsByAddress[account][i].isBurnAsset == filterBurned
            ) {
                count++;
            }
        }
        uint256[] memory indexes = new uint256[](count);
        uint256 index = 0;
        for (uint256 i = 0; i < _lockedAssetsByAddress[account].length; i++) {
            if (
                _lockedAssetsByAddress[account][i].amount > 0 &&
                _lockedAssetsByAddress[account][i].isBurnAsset == filterBurned
            ) {
                indexes[index] = i;
                index++;
            }
        }
        return indexes;
    }

    function getLockedAssetsCountByAccount(address account)
        public
        view
        returns (uint256)
    {
        uint256 count = 0;
        for (uint256 i = 0; i < _lockedAssetsByAddress[account].length; i++) {
            if (
                _lockedAssetsByAddress[account][i].account == account &&
                _lockedAssetsByAddress[account][i].amount > 0
            ) {
                count++;
            }
        }
        return count;
    }

    function claimableFees(address account, uint256 index)
        public
        view
        returns (uint256)
    {
        uint256 totalFees = 0;
        LockedAsset memory asset = _lockedAssetsByAddress[account][index];
        if (
            _poolFees.amount <= 0 ||
            asset.fees > _poolFees.amount ||
            asset.account != account ||
            asset.amount <= 0
        ) return 0;

        uint256 contractBalance = _poolFees.totalAssetAmount;
        if (asset.fees > _poolFees.amount) return 0;
        uint256 feesSinceDeposit =
            _feesAtIndex[_depositIndex].sub(_feesAtIndex[asset.depositIndex]); // gas saver
        uint256 reducedEnterPoolFees = asset.enterPoolFees;
        if (feesSinceDeposit > reducedEnterPoolFees) {
            reducedEnterPoolFees = 0;
        } else {
            reducedEnterPoolFees = reducedEnterPoolFees.sub(feesSinceDeposit);
        }
        
        uint256 fees = 0;
        if(_poolFees.amount >= reducedEnterPoolFees) {
            fees = _poolFees.amount.sub(reducedEnterPoolFees);
        }
        else {
            fees = _poolFees.amount;
        }


        uint256 senderBalance = asset.amount;
        uint256 weightPercent =
            senderBalance.mul(10000).div(contractBalance).mul(100);
        uint256 reward = fees.mul(weightPercent).div(1000000);
        totalFees = totalFees.add(reward);
        return totalFees;
    }

    function weightInPool(address account, uint256 index)
        public
        view
        returns (uint256)
    {
        LockedAsset memory asset = _lockedAssetsByAddress[account][index];
        uint256 contractBalance = _poolFees.totalAssetAmount;
        uint256 senderBalance = asset.amount;
        uint256 weightPercent =
            senderBalance.mul(10000).div(contractBalance).mul(100);
        return weightPercent;
    }

    function _deposit(
        address sender,
        uint256 amount,
        address referralAddr
    ) internal {
        require(
            _token.balanceOf(sender) >= amount,
            "Transfer amount exceeds balance"
        );
        require(
            _token.transferFrom(msg.sender, address(this), amount) == true,
            "Error transferFrom on the contract"
        );
        uint256 fee = (amount * _fee) / 10000;
        if (referralAddr != address(0) && referralAddr != msg.sender) {
            uint256 referralFee = (fee * 5000) / 10000;
            uint256 baseFee = fee;
            fee = fee.sub(referralFee);
            require(
                _token.transfer(referralAddr, referralFee) == true,
                "Error transferFrom on the contract for referral"
            );
            emit ReferralReward(
                sender,
                referralAddr,
                amount,
                referralFee,
                baseFee
            );
        }
        uint256 depositFee = (amount * _depositRewardFee) / 10000;
        uint256 amountSubFee = amount.sub(fee).sub(depositFee);

        // Add fees to withdrawable amount by the owner
        _feesVault = _feesVault.add(fee);
        if (_depositIndex == 0) {
            // If is the first deposit all fees goes the vault
            _feesVault = _feesVault.add(depositFee);
        } else {
            _poolFees.totalFeesAmount = _poolFees.totalFeesAmount.add(
                depositFee
            );
            _poolFees.amount = _poolFees.amount.add(depositFee);
        }
        _poolFees.totalAssetAmount = _poolFees.totalAssetAmount.add(
            amountSubFee
        );

        _depositIndex = _depositIndex.add(1);
        _poolFees.depositIndex = _depositIndex;
        _feesAtIndex[_depositIndex] = _poolFees.totalFeesAmount;

        _lockedAssetsByAddress[sender].push(
            LockedAsset({
                enterPoolFees: _poolFees.amount,
                account: sender,
                fees: depositFee,
                amount: amountSubFee,
                burnedHorn: 0,
                fromDate: block.timestamp,
                toDate: block.timestamp.add(_minLockDays * 1 days),
                alreadyClaimedHorn: 0,
                isBurnAsset: false,
                depositIndex: _depositIndex
            })
        );

        emit Deposit(
            sender,
            amount,
            fee.add(depositFee),
            block.timestamp,
            block.timestamp.add(_minLockDays * 1 days)
        );
    }

    function withdraw() external payable override returns (bool) {
        require(!_isPaused, "Withdraw are paused for now");
        _withdraw(msg.sender);
        return true;
    }

    function _withdraw(address sender) internal {
        uint256 totalWithdraw = 0;
        uint256 totalRemovedAssetsCount = 0;
        for (uint256 i = 0; i < _lockedAssetsByAddress[sender].length; i++) {
            LockedAsset storage asset = _lockedAssetsByAddress[sender][i];
            if (asset.toDate <= block.timestamp && asset.amount > 0) {
                totalRemovedAssetsCount = totalRemovedAssetsCount + 1;
                uint256 hornreward = 0;
                if (!asset.isBurnAsset) {
                    hornreward = _rewardWithdrawalHorn(
                        sender,
                        asset.amount,
                        asset.fromDate,
                        block.timestamp,
                        asset.alreadyClaimedHorn
                    );
                }
                uint256 reward = _claimFeesForAsset(asset, asset.amount);

                if (!asset.isBurnAsset) {
                    totalWithdraw = totalWithdraw.add(asset.amount).add(reward);
                } else {
                    totalWithdraw = totalWithdraw.add(reward);
                }
                asset.alreadyClaimedHorn = asset.alreadyClaimedHorn.add(
                    hornreward
                );
            }
        }

        // Reset amount
        LockedAsset[] storage newLockedAssetArray;
        for (uint256 i = 0; i < _lockedAssetsByAddress[sender].length; i++) {
            LockedAsset storage asset = _lockedAssetsByAddress[sender][i];
            if (asset.toDate <= block.timestamp && asset.amount > 0) {
                asset.amount = 0;
                asset.burnedHorn = 0;
            } else {
                newLockedAssetArray.push(_lockedAssetsByAddress[sender][i]);
            }
        }
        _lockedAssetsByAddress[sender] = newLockedAssetArray;

        if (totalWithdraw <= 0) return;
        require(
            _token.transfer(sender, totalWithdraw) == true,
            "Error transferFrom on the contract"
        );
        emit Withdraw(sender, totalWithdraw);
    }

    function burn(uint256 amount) external payable override returns (bool) {
        require(!_isPaused, "Burn are paused for now");
        require(amount > 0, "Invalid amount");
        _burn(msg.sender, amount);
        return true;
    }

    function _burn(address sender, uint256 burnAmount) internal {
        require(
            _hornToken.balanceOf(sender) >= burnAmount,
            "Transfer amount exceeds balance"
        );
        _hornToken.burn(sender, burnAmount);

        _depositIndex = _depositIndex.add(1);
        _lockedAssetsByAddress[sender].push(
            LockedAsset({
                enterPoolFees: _poolFees.amount,
                account: sender,
                fees: 0,
                amount: burnAmount.mul(_weightPerHorn).div(1 ether),
                burnedHorn: burnAmount,
                fromDate: block.timestamp,
                toDate: block.timestamp.add(_minLockDays * 1 days),
                alreadyClaimedHorn: 0,
                isBurnAsset: true,
                depositIndex: _depositIndex
            })
        );
        emit Burn(
            sender,
            burnAmount,
            burnAmount.mul(_weightPerHorn).div(1 ether)
        );
    }

    function claimHorn() public payable {
        require(!_hornRewardDisabled, "Claim are disabled for now");
        uint256 totalClaim = 0;
        for (
            uint256 i = 0;
            i < _lockedAssetsByAddress[msg.sender].length;
            i += 1
        ) {
            LockedAsset storage asset = _lockedAssetsByAddress[msg.sender][i];
            if (asset.amount > 0 && !asset.isBurnAsset) {
                uint256 reward =
                    _rewardWithdrawalHorn(
                        msg.sender,
                        asset.amount,
                        asset.fromDate,
                        block.timestamp,
                        asset.alreadyClaimedHorn
                    );
                asset.alreadyClaimedHorn = asset.alreadyClaimedHorn.add(reward);
                totalClaim = totalClaim.add(reward);
            }
        }
        if (totalClaim <= 0) return;
        emit Claim(msg.sender, totalClaim);
    }

    function _rewardWithdrawalHorn(
        address receiver,
        uint256 amount,
        uint256 fromDate,
        uint256 toDate,
        uint256 alreadyClaimed
    ) internal returns (uint256) {
        if (_hornRewardDisabled) return 0;

        uint256 period = toDate.sub(fromDate).div(60).div(60).div(24);
        if (period <= 0) return 0;
        uint256 reward =
            _hornPerDay.mul(amount.mul(100)).div(10000).mul(period);
        reward = reward.div(100).sub(alreadyClaimed);
        if (reward <= 0) return 0;
        _hornToken.mint(address(this), reward);
        require(
            _hornToken.transfer(receiver, reward) == true,
            "Error transferFrom for reward on the contract"
        );
        return reward;
    }

    function _claimFeesForAsset(
        LockedAsset storage asset,
        uint256 withdrawAmount
    ) internal returns (uint256) {
        if (_poolFees.amount <= 0 || asset.fees > _poolFees.amount) return 0;
        uint256 contractBalance = _poolFees.totalAssetAmount;
        uint256 feesSinceDeposit =
            _feesAtIndex[_depositIndex].sub(_feesAtIndex[asset.depositIndex]); // gas saver
        uint256 reducedEnterPoolFees = asset.enterPoolFees;
        if (feesSinceDeposit > reducedEnterPoolFees) {
            reducedEnterPoolFees = 0;
        } else {
            reducedEnterPoolFees = reducedEnterPoolFees.sub(feesSinceDeposit);
        }

        uint256 fees = 0;
        if(_poolFees.amount >= reducedEnterPoolFees) {
            fees = _poolFees.amount.sub(reducedEnterPoolFees);
        }
        else {
            fees = _poolFees.amount;
        }

        uint256 senderBalance = withdrawAmount;
        uint256 reward = (fees * (senderBalance.mul(10000).div(contractBalance).mul(100))) / 1000000;

        _poolFees.amount = _poolFees.amount.sub(reward);
        _poolFees.totalAssetAmount = _poolFees.totalAssetAmount.sub(
            withdrawAmount
        );
        return reward;
    }

    function withdrawFees() public payable {
        require(
            msg.sender == _owner || msg.sender == _createdFrom,
            "Only the owner can do this"
        );
        require(
            _token.transfer(msg.sender, _feesVault) == true,
            "Error transferFrom on the contract"
        );
        _feesVault = 0;
    }
}

// 
/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        // solhint-disable-next-line no-inline-assembly
        assembly { size := extcodesize(account) }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{ value: amount }("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
      return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value, string memory errorMessage) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.call{ value: value }(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data, string memory errorMessage) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.staticcall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    function _verifyCallResult(bool success, bytes memory returndata, string memory errorMessage) private pure returns(bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                // solhint-disable-next-line no-inline-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// 
/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address payable) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}

// 
/**
 * @dev Contract module that allows children to implement role-based access
 * control mechanisms.
 *
 * Roles are referred to by their `bytes32` identifier. These should be exposed
 * in the external API and be unique. The best way to achieve this is by
 * using `public constant` hash digests:
 *
 * ```
 * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
 * ```
 *
 * Roles can be used to represent a set of permissions. To restrict access to a
 * function call, use {hasRole}:
 *
 * ```
 * function foo() public {
 *     require(hasRole(MY_ROLE, msg.sender));
 *     ...
 * }
 * ```
 *
 * Roles can be granted and revoked dynamically via the {grantRole} and
 * {revokeRole} functions. Each role has an associated admin role, and only
 * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
 *
 * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
 * that only accounts with this role will be able to grant or revoke other
 * roles. More complex role relationships can be created by using
 * {_setRoleAdmin}.
 *
 * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
 * grant and revoke this role. Extra precautions should be taken to secure
 * accounts that have been granted it.
 */
abstract contract AccessControl is Context {
    using EnumerableSet for EnumerableSet.AddressSet;
    using Address for address;

    struct RoleData {
        EnumerableSet.AddressSet members;
        bytes32 adminRole;
    }

    mapping (bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) public view returns (bool) {
        return _roles[role].members.contains(account);
    }

    /**
     * @dev Returns the number of accounts that have `role`. Can be used
     * together with {getRoleMember} to enumerate all bearers of a role.
     */
    function getRoleMemberCount(bytes32 role) public view returns (uint256) {
        return _roles[role].members.length();
    }

    /**
     * @dev Returns one of the accounts that have `role`. `index` must be a
     * value between 0 and {getRoleMemberCount}, non-inclusive.
     *
     * Role bearers are not sorted in any particular way, and their ordering may
     * change at any point.
     *
     * WARNING: When using {getRoleMember} and {getRoleMemberCount}, make sure
     * you perform all queries on the same block. See the following
     * https://forum.openzeppelin.com/t/iterating-over-elements-on-enumerableset-in-openzeppelin-contracts/2296[forum post]
     * for more information.
     */
    function getRoleMember(bytes32 role, uint256 index) public view returns (address) {
        return _roles[role].members.at(index);
    }

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) public view returns (bytes32) {
        return _roles[role].adminRole;
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) public virtual {
        require(hasRole(_roles[role].adminRole, _msgSender()), "AccessControl: sender must be an admin to grant");

        _grantRole(role, account);
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) public virtual {
        require(hasRole(_roles[role].adminRole, _msgSender()), "AccessControl: sender must be an admin to revoke");

        _revokeRole(role, account);
    }

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) public virtual {
        require(account == _msgSender(), "AccessControl: can only renounce roles for self");

        _revokeRole(role, account);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event. Note that unlike {grantRole}, this function doesn't perform any
     * checks on the calling account.
     *
     * [WARNING]
     * ====
     * This function should only be called from the constructor when setting
     * up the initial roles for the system.
     *
     * Using this function in any other way is effectively circumventing the admin
     * system imposed by {AccessControl}.
     * ====
     */
    function _setupRole(bytes32 role, address account) internal virtual {
        _grantRole(role, account);
    }

    /**
     * @dev Sets `adminRole` as ``role``'s admin role.
     *
     * Emits a {RoleAdminChanged} event.
     */
    function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
        emit RoleAdminChanged(role, _roles[role].adminRole, adminRole);
        _roles[role].adminRole = adminRole;
    }

    function _grantRole(bytes32 role, address account) private {
        if (_roles[role].members.add(account)) {
            emit RoleGranted(role, account, _msgSender());
        }
    }

    function _revokeRole(bytes32 role, address account) private {
        if (_roles[role].members.remove(account)) {
            emit RoleRevoked(role, account, _msgSender());
        }
    }
}

// 
/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// 
/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.zeppelin.solutions/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * We have followed general OpenZeppelin guidelines: functions revert instead
 * of returning `false` on failure. This behavior is nonetheless conventional
 * and does not conflict with the expectations of ERC20 applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20 {
    using SafeMath for uint256;

    mapping (address => uint256) private _balances;

    mapping (address => mapping (address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;
    uint8 private _decimals;

    /**
     * @dev Sets the values for {name} and {symbol}, initializes {decimals} with
     * a default value of 18.
     *
     * To select a different value for {decimals}, use {_setupDecimals}.
     *
     * All three of these values are immutable: they can only be set once during
     * construction.
     */
    constructor (string memory name_, string memory symbol_) public {
        _name = name_;
        _symbol = symbol_;
        _decimals = 18;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5,05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the value {ERC20} uses, unless {_setupDecimals} is
     * called.
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view returns (uint8) {
        return _decimals;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `recipient` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address recipient, uint256 amount) public virtual override returns (bool) {
        _transfer(_msgSender(), recipient, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        _approve(_msgSender(), spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * Requirements:
     *
     * - `sender` and `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     * - the caller must have allowance for ``sender``'s tokens of at least
     * `amount`.
     */
    function transferFrom(address sender, address recipient, uint256 amount) public virtual override returns (bool) {
        _transfer(sender, recipient, amount);
        _approve(sender, _msgSender(), _allowances[sender][_msgSender()].sub(amount, "ERC20: transfer amount exceeds allowance"));
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        _approve(_msgSender(), spender, _allowances[_msgSender()][spender].add(addedValue));
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        _approve(_msgSender(), spender, _allowances[_msgSender()][spender].sub(subtractedValue, "ERC20: decreased allowance below zero"));
        return true;
    }

    /**
     * @dev Moves tokens `amount` from `sender` to `recipient`.
     *
     * This is internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `sender` cannot be the zero address.
     * - `recipient` cannot be the zero address.
     * - `sender` must have a balance of at least `amount`.
     */
    function _transfer(address sender, address recipient, uint256 amount) internal virtual {
        require(sender != address(0), "ERC20: transfer from the zero address");
        require(recipient != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(sender, recipient, amount);

        _balances[sender] = _balances[sender].sub(amount, "ERC20: transfer amount exceeds balance");
        _balances[recipient] = _balances[recipient].add(amount);
        emit Transfer(sender, recipient, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply = _totalSupply.add(amount);
        _balances[account] = _balances[account].add(amount);
        emit Transfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        _balances[account] = _balances[account].sub(amount, "ERC20: burn amount exceeds balance");
        _totalSupply = _totalSupply.sub(amount);
        emit Transfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(address owner, address spender, uint256 amount) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Sets {decimals} to a value other than the default one of 18.
     *
     * WARNING: This function should only be called from the constructor. Most
     * applications that interact with token contracts will not expect
     * {decimals} to ever change, and may work incorrectly if it does.
     */
    function _setupDecimals(uint8 decimals_) internal {
        _decimals = decimals_;
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be to transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 amount) internal virtual { }
}

// 
contract HornToken is ERC20, AccessControl {
    bytes32 public constant MINTER_ROLE = keccak256("MINTER_ROLE");
    bytes32 public constant BURNER_ROLE = keccak256("BURNER_ROLE");

    uint256 private _initialSupply = 5000000 * 10 ** 18;
    uint256 private _totalSupply = 0;
    uint256 private _alreadyMinted = 0;
    uint256 private _maxMintAmount = 5000000 * 10 ** 18;
    uint256 private _maxSupply = 10000000 * 10 ** 18;

    constructor() ERC20("Horn", "HORN") {
        _totalSupply = _initialSupply;
        _mint(msg.sender, _initialSupply);
        _setupRole(DEFAULT_ADMIN_ROLE, msg.sender);
        _setupRole(MINTER_ROLE, msg.sender);
        _setupRole(BURNER_ROLE, msg.sender);
    }

    function mint(address to, uint256 amount) public {
        require(hasRole(MINTER_ROLE, msg.sender), "Caller is not a minter");
        require(_totalSupply + amount <= _maxSupply, "Can't mint more tokens");
        require(_alreadyMinted + amount <= _maxMintAmount, "Can't mint more tokens");
        _totalSupply += amount;
        _alreadyMinted += amount;
        _mint(to, amount);
    }

    function burn(address from, uint256 amount) public {
        require(hasRole(BURNER_ROLE, msg.sender), "Caller is not a burner");
        _totalSupply -= amount;
        _burn(from, amount);
    }
}

// 
contract HornPreSale {
    uint256 public _hornPerEther = 5000000000000000000000;
    uint256 public _maxHornForPresale;
    uint256 public _hornSolded;
    bool public _paused = false;
    address private _owner;
    IExtendedERC20 private _hornToken;

    address internal _wethTokenAddress;

    constructor(address hornTokenAddr, uint256 maxHornForPresale, address wethTokenAddress, uint256 hornPerEther) {
        _owner = msg.sender;
        _hornSolded = 0;
        _hornPerEther = hornPerEther;
        _wethTokenAddress = wethTokenAddress;
        _maxHornForPresale = maxHornForPresale;
        _hornToken = IExtendedERC20(hornTokenAddr);
    }

    modifier onlyOwner() {
        require(msg.sender == _owner, "Only the owner can do this action");
        _;
    }

    function setMaxHornForPresale(uint256 value) public payable onlyOwner {
        _maxHornForPresale = value;
    }

    function setHornPerEther(uint256 value) public payable onlyOwner {
        _hornPerEther = value;
    }

    function setPaused(bool value) public payable onlyOwner {
        _paused = value;
    }
    
    function withdraw() public payable onlyOwner {
        IERC20 wethToken = IERC20(_wethTokenAddress);
        wethToken.transfer(msg.sender, wethToken.balanceOf(address(this)));
        msg.sender.transfer(address(this).balance);
    }

    receive() external payable {
        require(_paused == false, "Presale is paused");
        _mintAndTransferHorn(msg.value, msg.sender);
    }

    function transferToken(uint256 _amount) public payable {
        require(_paused == false, "Presale is paused");
        IERC20 token = IERC20(_wethTokenAddress);

        // Remove wether from the sender
        require(
            _amount > 0,
            "Invalid amount"
        );
        require(
            token.balanceOf(msg.sender) >= _amount,
            "No WETH amount required"
        );
        require(
            token.transferFrom(msg.sender, address(this), _amount) == true,
            "Can't transfert WETH on the contract"
        );

        _mintAndTransferHorn(_amount, msg.sender);
    }

    function _mintAndTransferHorn(uint256 inAmount, address sender) internal {
        uint256 hornAmount = inAmount * _hornPerEther;
        require(_hornSolded + hornAmount <= _maxHornForPresale, "The limit of the presale has been reached, try buy less horn");
        _hornToken.mint(address(this), hornAmount);
        _hornSolded = _hornSolded + hornAmount;
        require(
            _hornToken.transfer(sender, hornAmount) == true,
            "Error transferFrom on the contract"
        );
    }
}